Polarity preserving decon in "N log N" time

Jon Claerbout

ABSTRACT

A slight modification to Fourier spectral factorization enables deconvolution to preserve and enhance seismogram polarities. It spikes the center lobe of the Ricker wavelet. It works by tapering at small lags the antisymmetric part of the time-domain representation of the log spectrum.

INTRODUCTION

On a microscopic scale, the arrival of seismic events is always emergent, consequently use of strict causality in mathematical analysis often leads to disappointing results. Here we see how a slight modification to Fourier spectral factorization allowing slight noncausality (half period) easily handles the Ricker wavelet, a commonly observed emergent waveform.

Conflicting goals

Two goals of seismogram source waveform estimation conflict. They are:

1. Preserve and clearly exhibit the polarity of seismic reflections.
2. Estimate and use for data processing a source waveform that is causal, namely, the response vanishes before the excitation.

When we honor one, we find trouble with the other. This conflict will be defined and resolved here.

Prevalence of Ricker wavelet

The conflict is most directly seen and addressed in the specific case of the Ricker wavelet. It is generally seen on the water bottom and on any strong reflector such as the top of salt and often the bottom of salt. The Ricker wavelet obscures recognition of multiple reflections by their polarity alternations. (For example, look ahead to the left side of Figure 2.)

Definition of Ricker wavelet

We use the term "Ricker wavelet" to describe any wave shape resembling the second derivative of a Gaussian. The reason Ricker wavelets are so prevalent in marine seismology is this: The water surface negative reflection coefficient causes two ghosts, one at the gun, the other at the receiver. At each location the slightly delayed surface reflection (of negative polarity) applies a finite difference to the image. Consequently, any impulsive reflector looks like a band-limited second derivative. Land seismometers measure neither displacement nor velocity. Essentially, they are devices that measure acceleration. Band limiting their output again produces wavelets like the Ricker wavelet.

Symmetrical decon preserves polarity. Hooray!

The most primative deconvolution is a symmetrical filter, for example $\mathrm{FT}^{-1}(1/|D(\omega)|)$ where $|D(\omega)|$ denotes the amplitude spectrum of the data. The output $\mathrm{FT}^{-1}(D/|D|)$ of this filter is spectrally white. This decon filter converts a Ricker wavelet to an impulse at its center. Hooray! Notice this impulse will have the opposite polarity of that of predictive decon which attempts to spike the first lobe of the Ricker wavelet.



Figure 1: The non-causality needed is defined by the small backward distance from the center of the Ricker wavelet to its onset. [**NR**]

How symmetric decon fails

Symmetric decon is wonderful the way it preserves polarity, but it has one feature that is really embarassing. About 150ms after the air gun blast is a bubble collapse blast. At early times they are quite different in frequency content and size, but because the bubble contains mostly the lower frequencies, its relative contribution becomes much stronger later in the record. What symmetric decon does with this bubble is horrifying. It gives a precursor 150ms before the water bottom. And also before every other event!

How predictive decon fails

Traditional predictive decon attempts to convert the Ricker wavelet to an impulse at its onset. This works badly because the onset time of the emergent signal, the Ricker wavelet, is not well defined.

GETTING THE BEST OF BOTH WORLDS

The good news is that it is possible to have the best of both worlds, both to deal time-symmetrically with the Ricker wavelet and non-time-symmetrically with the bubble. Furthermore, we do it rapidly in fast Fourier $N \log_2 N$ time. The Ricker wavelet is quite short, about 15ms, while the bubble delay is quite long, about 150ms, so the separation of the two is not delicate.

The less happy news is that the required theory is quite deep and not well known. It was first invented by famous mathematicians about 1940, and did not reach text-book status until my 1974 book FGDP, with code included in my newer book PVI 1992. Luckily, I have been teaching this material for many years now and believe I can extract the essence without great pain for you.

Spectral factorization

A causal function is one that vanishes at negative time. Too short a summary is to say the exponential of a causal is a causal. What is meant is if we take the Fourier transformation of a causal function, exponentiate it, and then inverse transform we will again have a causal function. This is the heart of spectral factorization, an obscure mathematical calculation addressing interesting practical applications.

Start with Z-transforms. Given a time function $(1, u_1, u_2, u_3, \cdots)$ its Z-transform is $U(Z) = 1 + u_1 Z + u_2 Z^2 + u_3 Z^3 + \cdots$. When you identify $Z = e^{i\omega\Delta t}$ and $Z^5 = e^{i\omega 5\Delta t}$ the Z-transform is clearly a Fourier series. An example of a causal function is u_τ. It is causal because $u_\tau = 0$ for $\tau < 0$ likewise, $U(Z)$ has no powers of $1/Z$.

We may exponentiate $U(Z)$ by a frequency domain method or a time domain method. Easiest is the frequency domain method. Write $e^{U(Z(\omega))}$ for all ω, then Fourier transform to time. More interesting is the time domain method. The polynomial U has no powers of $1/Z$. The power series for an exponential is $e^U = 1 + U + U^2/2! + U^3/3! + \cdots$. Inserting the polynomial for U into the power series for e^U gives us a new polynomial (infinite series) that has no powers of $1/Z$. Furthermore, this new polynomial always converges because of the powerful influence of the denominator factorials. Thus we have shown that the "exponential of a causal is a causal".

Let $\bar{S}(Z(\omega))$ be an amplitude spectrum $\bar{S}(\omega) > 0$ with logarithm $\bar{U} = \log \bar{S}$. The exponential is the inverse of the logarithm

$$\bar{S} \;=\; e^{\log \bar{S}} \;=\; e^{\bar{U}} \tag{1}$$

Both $\bar{S}$ and $\bar{U}$ are real symmetric functions of ω. In the time domain, $|\bar{S}|^2$ corresponds to an autocorrelation. In the time domain, $\bar{U}$ merely corresponds to a real symmetric function $\bar{u}_\tau$. Adding some phase function $\Phi(\omega)$ to $\bar{U}$ will shift the time function s_τ, likely shifting each frequency differently.

$$S \;=\; e^{\log \bar{S}+i\Phi} \;=\; e^{\bar{U}+i\Phi} \;=\; e^{U} \tag{2}$$

Keeping $\bar{U}$ fixed keeps the spectrum S^*S fixed. Let u_τ now correspond to the Fourier transform of $U(\omega) = \bar{U} + i\Phi$. The time symmetric part of u_τ corresponds to $\bar{U}(\omega)$ while the antisymmetric part of u_τ corresponds to the newly added phase $\Phi(\omega)$. How shall we choose $\Phi(\omega)$? Let us choose the antisymmetric part of u_τ instead, choose it to cancel the symmetric part of u_τ on the negative τ axis. In other words, let us choose u_τ to be causal. Recalling that "exponentials of causals are causal" we have thus created a causal s_τ. Hooray! Hooray because s_τ has the same spectrum $\bar{S}$ that we started with. We started with a spectrum $\bar{S}$ and constructed a causal wavelet s_τ with that spectrum. Good trick! This is called "spectral factorization." Causal decon is simply taking your data D and dividing by a causal source waveform S.

Mostly causal decon

Now for the innovation. There are many pitfalls in the log domain. Seismologists are accustomed to ignoring the scale of their signals. Let the plot program figure out a suitable scale, we think. Once you take the logarithm of a signal, you are in a different world. If you double the log, you have squared the original signal. Got to be careful! What you can do safely with log signals is add or subtract something. This has the effect of scaling the original signal. Create an anticausal function $u_\tau^{\text{anti}} = \text{sgn}(\tau) u_\tau$. The signum function $\text{sgn}(\tau)$ is -1 for $\tau < 0$ and $+1$ for $\tau > 0$. Adding this anticausal function to u_τ zeros the negative lags while doubling the positive lags. Because it is antisymmetric it changes the phase spectrum. It does not change the amplitude spectrum. We can use any anti-symmetrical function we wish to monkey with the phase while not changing the amplitude. We could add the antisymmetric function u_τ^{anti}, but that would simply do traditional causal decon. Instead, near the origin we taper u_τ^{anti} towards zero. This creates symmetric Ricker-like behavior near the origin while leaving causal behavior further away. The tapering zone used here extends beyond the Ricker width, about 20ms, but not so far as the bubble delay, about 150ms, an easy distinction. A parallel analysis is found in another paper in this report (Claerbout et al. (2012)).

It's easy. Figures 2 and Figure 3 show the desired behavior. Hooray! The results are lovely. Better yet, they are not the end but the beginning. They are based on the simple notion that we want a white output spectrum. Our real goal is a sparse time function, not a white one. The results in these figures are simply the starting point of another paper in this report.



Figure 2: Left is Yilmaz and Cumro shot profile 33. Right is the result of "polarity preserving deconvolution." Observe enhanced visibility of alternating polarity of multiple reflections. [**ER**]



Figure 3: Shot waveform extracted from a constant offset section from the Gulf of Mexico. Peaks at zero lag as does symmetric Ricker wavelet, but for unknown reasons the peak heights are not in the (-1,2,-1) proportions of a Ricker wavelet. [**ER**]

DISCUSSION AND CONCLUSION

This paper introduces the notion that by manipulating the u_τ we may make improvements on the old mathematical method of blind deconvolution. We were uncommonly successful here in dealing with our most commonly observed wavelet, the Ricker wavelet. This success suggests other improvements might flow from manipulations of the u_τ for other purposes.

For example, given only a single seismogram, we may wish to limit the number of degrees of freedom for the filter estimation. We have long known this can be done by smoothing the data spectrum. Another method is to limit the range, or taper the range of u_τ coefficients. Such ideas are untried, so not yet compared.

Likewise, many shot waveforms have been recorded and tabulated. Perhaps it makes sense to map these wavelets to the "lag-log" space u_τ to better understand their statistics.

I see no immediate application, but we might recall that spectral factorization is also applicable for complex-valued signals. Then the spectrum is non-symmetric. This arises when time-dependent signals have been previously Fourier transformed over space.

Shuki asks, "What about seafloor receivers where there is one ghost, not two?" I reply, "Perhaps the same code can be used, but instead of gateing on the range $\pm\tau$ being 3/4 period for the Ricker wavelet, it might be instead 1/4 period for the the primary lobe.

APPENDIX

Subroutine `ftu` below is an ancient FT program from my book FGDP with conventional scaling consistent with Z-transforms. Data length must be a power of two. Subroutine `kolmogoroff` below was taken from my book PVI, converted from energy spectra to amplitude spectra. An insert in the middle implements the innovation of this paper; it diminishes the asymmetric part of u_τ near $|\tau| = 0$. A cosine squared weight was arbitrarily chosen. The suppression range was chosen from the origin to half way to an expected bubble on 4ms data.

While looking at the code you might notice that you also have the ability to taper large lags to shorten your filter response. This might be useful when you want to crop off downgoing multiples from your source waveform. It would also be helpful when you have insufficient data to be estimating long source waveforms.

```
subroutine kolmogoroff( n, cx)  # Spectral factorization.
integer i,              n       # input:  cx = amplitude spectrum
complex cx(n)                   # output: cx = FT of min phase wavelet
integer lag
```

```
real weight, asym
do i= 1, n
        cx(i) = clog( cx(i) )
call ftu( -1., n, cx)
do i= 2, n/2 {
        cx(i)     = cx(i) * 2.
        cx(n-i+2) = 0.
        }

# BEGIN stuff added to remove a little of the asymmetric part.
lag = 15        #   lag = 60ms/4ms where 60ms is half way to bubble.
do i = 2, lag {
        asym = (cx(i) - cx(n-i+2))/2.
        weight = cos( .5* 3.1416 * (i-1.)/(lag-1.))**2
        cx(i)     =  cx(i)     - weight * asym
        cx(n-i+2) =  cx(n-i+2) + weight * asym
        }
# END stuff added to remove a little of the asymmetric part.

call ftu( +1., n, cx)
do i= 1, n
        cx(i) = cexp( cx(i))
return; end


subroutine ftu( signi, nx, cx )
#   complex fourier transform with traditional scaling
#
#               1         nx          signi*2*pi*i*(j-1)*(k-1)/nx
#   cx(k)  =  -------- * sum cx(j) * e
#             scale      j=1             for k=1,2,...,nx=2**integer
#
#  scale=1 for forward transform signi=1, otherwise scale=1/nx
integer nx, i, j, k, m, istep, pad2
real    signi, arg
complex cx(nx), cmplx, cw, cdel, ct
do i= 1, nx
        if( signi<0.)
                cx(i) = cx(i) / nx
j = 1;  k = 1
do i= 1, nx {
        if (i<=j) { ct = cx(j); cx(j) = cx(i); cx(i) = ct }
        m = nx/2
        while (j>m && m>1) { j = j-m; m = m/2 }         # "&&" means .AND.
        j = j+m
```

```
        }
repeat {
        istep = 2*k;   cw = 1.;   arg = signi*3.14159265/k
        cdel = cmplx( cos(arg), sin(arg))
        do m= 1, k {
                do i= m, nx, istep
                    { ct=cw*cx(i+k);  cx(i+k)=cx(i)-ct;  cx(i)=cx(i)+ct}
                cw = cw * cdel
                }
        k = istep
        if(k>=nx) break
        }
return; end
```

REFERENCES

Claerbout, J., A. Guitton, and Q. Fu, 2012, Decon in the log domain with variable gain: SEP-Report, **147**, 313–322.